WS - Project 1
Football Data Web Service

Alexandre Ribeiro 108122
Guilherme Amorim 107162
Paulo Macedo 102620

March 2025

1 Introduction

Football is the most watched sport in the world, with millions of players, fans, and clubs spread across continents [1]. This sport generates a huge amount of data, from player statistics to club performance. Analyzing and visualizing these data can provide valuable insights into team interactions, player relationships, and overall trends in the football system.

In this project, available Kaggle football datasets were used to construct a semantic network reflecting relations among football clubs and players with stats and other interesting entities involved, such as football leagues and nations. The process involves structuring data into an interactive network, in which users can query relations in the football world through a web application.

2 Data

2.1 Sources

To gather the necessary information, multiple data sources were used :

- **Hubert Sidorowicz's Football Players Stats (2024-2025)**[**2**]: This dataset was taken from Kaggle and contains the essential information regarding the players, their statistics and their affiliation with different clubs across the 2024-2025 season (for the Big 5 European leagues only).
- **excel4soccer's ESPN Soccer Data** [**3**]: These datasets were also taken from Kaggle and contain additional information that complements the previous data: one dataset shows many clubs and their colors and logo; the other one relates a team with a city and a stadium.
- **zerozero.pt Scraping** [**4**]: In order to complete our semantic network, a dataset containing players photos urls was created through scraping a football website.

2.2 Transformations

The datasets were converted from CSV to RDF/NT/N3 format using a Python script that employed the rdflib and pandas libraries. The libraries simplify the conversion by offering simple syntax for handling RDF concepts such as namespaces, triples, types and much more.

2.2.1 Hubert Sidorowicz's Football Players Stats (2024-2025)

The dataset's creator provides two versions: a full version with 2,752 player records across 267 columns and a light version with the same number of rows but only 165 columns. For this work, the lighter version was selected and further refined to include only the most relevant attributes.

The selected attributes fall into three main categories:

- **Player Information**: Includes details such as name, positions, country of origin, birth year, and relationships with current and past clubs.
- **Player Statistics**: Comprises various performance metrics categorized under 'Attacking,' 'Defending,' and 'Goalkeeping,' with examples like 'Goals,' 'Tackles,' and 'Saves.'
- **Country and League Entities**: Contains essential details for building country and league entities, including abbreviations, names, and flags.

2.2.2 excel4soccer's ESPN Soccer Data

This dataset was used to supplement the previous one by providing additional details about each player's club. While the dataset's creator offers a broad collection of CSV files covering various aspects of football, only two — teams.csv and venues.csv — were utilized. These files contained essential information for constructing club entities, including club colors and logo and location information.

2.2.3 zerozero.pt Scraping

Since neither of the previous datasets included a complete set of player headshots, a web scraping script was developed to locate each player's zerozero.pt profile and extract the URL of their photo.

2.3 Result

By gathering information from the various datasets, the following semantic network was built:



Figure 1: Semantic Network Model

3 Queries

This section presents the queries used to extract information from the generated RDF dataset for the web application. These queries fall into three main categories: **player**, **club**, and **graph**. Queries labeled with the "Example" tag indicate that the subject (such as a player, club, or statistic) is dynamic and replaced with the relevant entity during execution in the web application. Not all queries are listed, only the most important ones.

3.1 Player

3.1.1 List All Players

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX fut-rel: <http://football.org/rel/>

SELECT
    ?player_id
    ?name
    (GROUP_CONCAT(DISTINCT ?position; separator=", ") AS ?positions)
    ?nation
    ?flag
    ?currentClub
    ?currentClubLogo
    ?born
WHERE {
    ?player_id rdf:type fut-rel:Player ;
               fut-rel:name ?name ;
               fut-rel:position ?position ;
               fut-rel:nation [ fut-rel:abrv ?nation ; fut-rel:flag ?flag ] ;
               fut-rel:born ?born ;
               fut-rel:club ?currentClub .

    OPTIONAL { ?currentClub fut-rel:logo ?currentClubLogo . }
}
GROUP BY ?player_id ?name ?nation ?flag ?born ?currentClub ?currentClubLogo
ORDER BY ?name
```

	player_id	name	positions	nation	flag	currentClub	currentClubLogo	born
1	http://football.org/ent/aaron_ciammaglichella	"Aaron Ciammaglichella"	"MF"	"ITA"	"https://a.espncdn.com/i/teamlogos/countries/500/ita.png"	http://football.org/ent/arsenal	"https://a.espncdn.com/i/teamlogos/soccer/500/359.png"	"2005"^^xsd:integer
2	http://football.org/ent/aaron_cresswell	"Aaron Cresswell"	"DF"	"ENG"	"https://a.espncdn.com/i/teamlogos/countries/500/eng.png"	http://football.org/ent/angers	"https://a.espncdn.com/i/teamlogos/soccer/500/7868.png"	"1989"^^xsd:integer
3	http://football.org/ent/aaron_malouda	"Aaron Malouda"	"FW"	"FRA"	"https://a.espncdn.com/i/teamlogos/countries/500/fra.png"	http://football.org/ent/lille	"https://a.espncdn.com/i/teamlogos/soccer/500/166.png"	"2005"^^xsd:integer
4	http://football.org/ent/	"Aaron Ramsdale"	"GK"	"ENG"	"https://a.espncdn.com/i/	http://football.org/ent/	"https://a.espncdn.com/i/	"1998"^^xsd:integer

Figure 2: List All Players (Output)

3.1.2 Get Player Details (Example)

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX fut-rel: <http://football.org/rel/>

SELECT
    ?player_id
    ?name
    (GROUP_CONCAT(DISTINCT ?position; separator=", ") AS ?positions)
    ?nation
    ?flag
    ?currentClub
    ?currentClubName
    ?currentClubLogo
    ?currentClubColor
    ?currentClubAltColor
    (GROUP_CONCAT(DISTINCT ?pastClub; separator=", ") AS ?pastClubs)
    (GROUP_CONCAT(DISTINCT ?pastClubName; separator=", ") AS ?pastClubNames)
    (GROUP_CONCAT(DISTINCT ?pastClubLogo; separator=", ") AS ?pastClubLogos)
    ?born
WHERE {
    # Filter for a specific player by ID
    VALUES ?player_id { <http://football.org/ent/marcus_rashford> }

    ?player_id rdf:type fut-rel:Player ;
               fut-rel:name ?name ;
               fut-rel:position ?position ;
               fut-rel:nation [ fut-rel:name ?nation ; fut-rel:flag ?flag ] ;
               fut-rel:born ?born ;
               fut-rel:club ?currentClub .

    ?currentClub fut-rel:name ?currentClubName ;
                 fut-rel:logo ?currentClubLogo ;
                 fut-rel:color ?currentClubColor ;
                 fut-rel:alternateColor ?currentClubAltColor .

    # Get past clubs
    OPTIONAL {
        ?player_id fut-rel:past_club ?pastClub .
        ?pastClub fut-rel:name ?pastClubName ;
                  fut-rel:logo ?pastClubLogo .
    }
}
GROUP BY ?player_id ?name ?nation ?flag ?currentClub ?currentClubName
         ?currentClubLogo ?currentClubColor ?currentClubAltColor ?born
```

	player_id	name	positions	nation	flag	currentClub	currentClubName	currentClubLogo	currentClubColor	currentClubAltColor	pastClubs	pastClubNames	pastClubLogos	born
1	http://football.org/ent/marcus_rashford	"Marcus Rashford"	"MF, FW"	"England"	"https://a.espncdn.com/i/teamlogos/countries/500/eng.png"	http://football.org/ent/aston_villa	"Aston Villa"	"https://a.espncdn.com/i/teamlogos/soccer/500/362.png"	"660e36"	"ffffff"	"http://football.org/ent/man_united"	"Manchester United"	"https://a.espncdn.com/i/teamlogos/soccer/500/360.png"	"1997"^^xsd:integer

Figure 3: Get Player Details (Output)

3.1.3 Get Player Statistics (Example)

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX fut-rel: <http://football.org/rel/>

SELECT ?stat_category ?stat_name ?stat_value
WHERE {
    VALUES ?player_id { <http://football.org/ent/bruno_fernandes> }

    ?player_id ?stat ?stat_value .
    ?stat fut-rel:type/fut-rel:name ?stat_category ;
          fut-rel:name ?stat_name .
}
```

	stat_category	stat_name	stat_value
1	"Passing & Creativity"	"Miscontrols"	"29"^^xsd:integer
2	"Defending"	"Errors Leading to Goal"	"2"^^xsd:integer
3	"Miscellaneous"	"Red Cards"	"2"^^xsd:integer
4	"Defending"	"Tackles Won"	"41"^^xsd:integer
5	"Defending"	"Clearances"	"30"^^xsd:integer
6	"Miscellaneous"	"Own Goals"	"0"^^xsd:integer

Figure 4: Get Player Statistics (Output)

3.1.4 Get Top 10 Players By Stat (Example)

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX fut-rel: <http://football.org/rel/>
PREFIX fut-stat: <http://football.org/stat/>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>

SELECT
    ?player_id
    ?name
    ?photo_url
    ?stat_name
    ?stat_value
    ?club_name
    ?club_logo
    ?color
    ?alternateColor
    ?flag
WHERE {
    VALUES ?stat { <http://football.org/stat/gls> }

    ?player_id rdf:type fut-rel:Player ;
               fut-rel:name ?name ;
               ?stat ?stat_value ;
               fut-rel:club ?club ;
               fut-rel:nation/fut-rel:flag ?flag ;
               fut-stat:min ?min .

    ?club fut-rel:name ?club_name ;
          fut-rel:logo ?club_logo ;
          fut-rel:color ?color ;
          fut-rel:alternateColor ?alternateColor .

    ?stat fut-rel:name ?stat_name .

    OPTIONAL { ?player_id fut-rel:photo_url ?photo_url . }
}
ORDER BY DESC(xsd:float(?stat_value)) ?min
LIMIT 10
```



	player_id	name	photo_url	stat_name	stat_value	club_name	club_logo	color	alternateColor	flag
1	http://football.org/ent/mohamed_salah	"Mohamed Salah"	"https://resources.premierleague.com/premierleague/photos/players/250x250/p118748.png"	"Goals"	"27"^^xsd:integer	"Liverpool"	"https://a.espncdn.com/i/teamlogos/soccer/500/364.png"	"d11317"	"ffffff"	"https://a.espncdn.com/i/teamlogos/countries/500/egy.png"
2	http://football.org/ent/mateo_retegui	"Mateo Retegui"	"https://www.zerozero.pt/img/jogadores/new/52/55/615255_mateo_retegui_20241229232146.png"	"Goals"	"22"^^xsd:integer	"Atalanta"	"https://a.espncdn.com/i/teamlogos/soccer/500/105.png"	"1157bf"	"ffffff"	"https://a.espncdn.com/i/teamlogos/countries/500/ita.png"
3	http://football.org/ent/robert_lewandowski	"Robert Lewandowski"	"https://www.zerozero.pt/img/jogadores/	"Goals"	"22"^^xsd:integer	"Barcelona"	"https://a.espncdn.com/i/teamlogos/soccer/	"a20046"	"2685de"	"https://a.espncdn.com/i/teamlogos/

Figure 5: Get Top 10 Players By Stat (Output)

3.1.5 Update Player Club (Example)

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX fut-rel: <http://football.org/rel/>

DELETE {
    <http://football.org/ent/mohamed_salah> fut-rel:club ?oldClub .
}
INSERT {
    <http://football.org/ent/mohamed_salah> fut-rel:club <http://football.org/ent/everton> .
    <http://football.org/ent/mohamed_salah> fut-rel:past_club <http://football.org/ent/liverpool> .
}
WHERE {
    OPTIONAL { <http://football.org/ent/mohamed_salah> fut-rel:club ?oldClub . }
}
```


Figure 6: Update Player Club (Output)

3.1.6 Create Player (Example)

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX fut-rel: <http://football.org/rel/>
PREFIX fut-stat: <http://football.org/stat/>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>

INSERT DATA {
    <http://football.org/ent/cristiano_ronaldo> rdf:type fut-rel:Player ;
        fut-rel:name "Cristiano Ronaldo";
        fut-rel:born 1985 ;
        fut-rel:position "FW" ;

        fut-rel:photo_url " https://www.zerozero.pt/img/jogadores/new/15/79/1579_cristiano_ronaldo_20250121144448.png" ;
        fut-rel:nation <http://football.org/ent/CHI> ;
        fut-rel:club <http://football.org/ent/bologna> ;

        fut-stat:mp 0 ;
        # For this example, the remaining stats are hidden
        fut-stat:off_stats_misc 0 .
}
```

Figure 7: Create Player (Output)

3.1.7 Delete Player (Example)

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX fut-rel: <http://football.org/rel/>
PREFIX fut-stat: <http://football.org/stat/>

DELETE {
    # Delete all properties where player is the subject
    <http://football.org/ent/mohamed_salah> ?p ?o .

    # Delete all properties where player is the object
    ?s ?p2 <http://football.org/ent/mohamed_salah> .
}
    WHERE {
    # Get all statements where player is the subject
    <http://football.org/ent/mohamed_salah> ?p ?o .

    # Get all statements where player is the object
    OPTIONAL {{ ?s ?p2 <http://football.org/ent/mohamed_salah> . }}
}
```


Figure 8: Delete Player (Output)

3.1.8 Player Connection (Same Country Example)

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX fut-rel: <http://football.org/rel/>

ASK {
    <http://football.org/ent/bruno_fernandes> fut-rel:nation ?country .
    <http://football.org/ent/goncalo_ramos> fut-rel:nation ?country .
}
```



Figure 9: Player Connection (Output)

3.2 Club

3.2.1 List All Clubs

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX fut-rel: <http://football.org/rel/>
```

```
SELECT
    ?club_id
    ?abbreviation
    ?name
    ?league
    ?flag
    ?logo
    ?color
    ?alternateColor
    (COUNT(?player) AS ?numPlayers)
WHERE {
    ?club_id rdf:type fut-rel:Club ;
             fut-rel:name ?name ;
             fut-rel:abrv ?abbreviation ;
             fut-rel:league/fut-rel:name ?league ;
             fut-rel:country/fut-rel:flag ?flag ;
             fut-rel:logo ?logo ;
             fut-rel:color ?color ;
             fut-rel:alternateColor ?alternateColor .

    OPTIONAL { ?player fut-rel:club ?club_id . }
}
GROUP BY ?club_id ?abbreviation ?name ?league ?flag ?logo ?color ?alternateColor
ORDER BY ?name
```

	club_id	abbreviation	name	league	flag	logo	color	alternateColor	numPlayers
1	http://football.org/ent/heidenheim	"HDH"	"1. FC Heidenheim 1846"	"Bundesliga"	"https://a.espncdn.com/i/teamlogos/countries/500/ger.png"	"https://a.espncdn.com/i/teamlogos/soccer/500/6418.png"	"DA0308"	"003399"	"25"^^xsd:integer
2	http://football.org/ent/union_berlin	"FCU"	"1. FC Union Berlin"	"Bundesliga"	"https://a.espncdn.com/i/teamlogos/countries/500/ger.png"	"https://a.espncdn.com/i/teamlogos/soccer/500/598.png"	"DA0308"	"d4d4d4"	"27"^^xsd:integer
3	http://football.org/ent/milan	"MIL"	"AC Milan"	"Serie A"	"https://a.espncdn.com/i/teamlogos/countries/500/ita.png"	"https://a.espncdn.com/i/teamlogos/soccer/500/103.png"	"e4002b"	"ffffff"	"33"^^xsd:integer
4	http://football.org/ent/bournemouth	"BOU"	"AFC Bournemouth"	"Premier League"	"https://a.espncdn.com/i/teamlogos/countries/500/eng.png"	"https://a.espncdn.com/i/teamlogos/soccer/500/349.png"	"f42727"	"ffffff"	"26"^^xsd:integer

Figure 10: List All Clubs (Output)

3.2.2 Get Club Details (Example)

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX fut-rel: <http://football.org/rel/>

SELECT
    ?club_id
    ?abbreviation
    ?name
    ?stadium
    ?city
    ?league_id
    ?league_name
    ?flag
    ?logo
    ?color
    ?alternateColor
WHERE {
    VALUES ?club_id { <http://football.org/ent/heidenheim> }

    ?club_id rdf:type fut-rel:Club ;
             fut-rel:name ?name ;
             fut-rel:abrv ?abbreviation ;
             fut-rel:stadium ?stadium ;
             fut-rel:city ?city ;
             fut-rel:league ?league_id ;
             fut-rel:logo ?logo ;
             fut-rel:color ?color ;
             fut-rel:alternateColor ?alternateColor ;
             fut-rel:country/fut-rel:flag ?flag .

    ?league_id fut-rel:name ?league_name .
}
```

	club_id	abbreviation	name	stadium	city	league_id	league_name	flag	logo	color	alternateColor
1	http://football.org/ent/heidenheim	"HDH"	"1. FC Heidenheim 1846"	"Voith-Arena"	"Heidenheim"	http://football.org/ent/de	"Bundesliga"	"https://a.espncdn.com/i/teamlogos/countries/500/ger.png"	"https://a.espncdn.com/i/teamlogos/soccer/500/6418.png"	"DA0308"	"003399"

Figure 11: Get Club Details (Output)

3.2.3 Get Club Statistics (Example)

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX fut-rel: <http://football.org/rel/>

SELECT ?stat_category ?stat_name ?stat_value
WHERE {
    VALUES ?club_id { <http://football.org/ent/heidenheim> }

    ?club_id ?stat ?stat_value .
    ?stat fut-rel:type [fut-rel:name ?stat_category];
          fut-rel:name ?stat_name .
}
```

	stat_category	stat_name	stat_value
1	"Passing & Creativity"	"Miscontrols"	"420"^^xsd:integer
2	"Defending"	"Errors Leading to Goal"	"9"^^xsd:integer
3	"Miscellaneous"	"Red Cards"	"1"^^xsd:integer
4	"Defending"	"Tackles Won"	"285"^^xsd:integer
5	"Defending"	"Clearances"	"741"^^xsd:integer
6	"Miscellaneous"	"Own Goals"	"0"^^xsd:integer
7	"Attacking"	"Penalties Scored"	"4"^^xsd:integer

Figure 12: Get Club Stats (Output)

3.2.4 Get Club Players (Example)

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX fut-rel: <http://football.org/rel/>

SELECT
    ?player_id
    ?name
    ?born
    (GROUP_CONCAT(DISTINCT ?position; separator=", ") AS ?positions)
    ?nation
    ?flag
WHERE {
    VALUES ?club_id { <http://football.org/ent/heidenheim> }

    ?player_id rdf:type fut-rel:Player ;
               fut-rel:name ?name ;
               fut-rel:born ?born ;
               fut-rel:club ?club_id ;
               fut-rel:nation/fut-rel:name ?nation ;
               fut-rel:nation/fut-rel:flag ?flag ;
               fut-rel:position ?position .
}
GROUP BY ?player_id ?name ?born ?nation ?flag
ORDER BY ?name
```

	player_id	name	born	positions	nation	flag
1	http://football.org/ent/adrian_beck	"Adrian Beck"	"1997"^^xsd:integer	"MF, FW"	"Germany"	"https://a.espncdn.com/i/teamlogos/countries/500/ger.png"
2	http://football.org/ent/benedikt_gimber	"Benedikt Gimber"	"1997"^^xsd:integer	"DF"	"Germany"	"https://a.espncdn.com/i/teamlogos/countries/500/ger.png"
3	http://football.org/ent/budu_zivzivadze	"Budu Zivzivadze"	"1994"^^xsd:integer	"FW"	"Georgia"	"https://a.espncdn.com/i/teamlogos/countries/500/geo.png"
4	http://football.org/ent/denis_thomalla	"Denis Thomalla"	"1992"^^xsd:integer	"MF"	"Germany"	"https://a.espncdn.com/i/teamlogos/countries/500/ger.png"
5	http://football.org/ent/jan_schoppner	"Jan Schöppner"	"1999"^^xsd:integer	"MF"	"Germany"	"https://a.espncdn.com/i/teamlogos/countries/500/ger.png"
6	http://football.org/ent/jonas_fohrenbach	"Jonas Föhrenbach"	"1996"^^xsd:integer	"DF"	"Germany"	"https://a.espncdn.com/i/teamlogos/countries/500/ger.png"

Figure 13: Get Club Players (Output)

3.2.5 Get Top 10 Clubs By Stat (Example)

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX fut-rel: <http://football.org/rel/>

SELECT
    ?club_id
    ?name
    ?stat_name
    ?stat_value
    ?logo
WHERE {
    VALUES ?stat { <http://football.org/stat/crdr> }

    ?club_id rdf:type fut-rel:Club ;
                    fut-rel:name ?name ;
                       fut-rel:logo ?logo ;
             ?stat ?stat_value .
    ?stat fut-rel:name ?stat_name .
}
ORDER BY DESC(?stat_value) ?club_id
LIMIT 10
```

	club_id	name	stat_name	stat_value	logo
1	http://football.org/ent/verona	"Hellas Verona"	"Red Cards"	"8"^^xsd:integer	"https://a.espncdn.com/i/teamlogos/soccer/500/119.png"
2	http://football.org/ent/lens	"Lens"	"Red Cards"	"7"^^xsd:integer	"https://a.espncdn.com/i/teamlogos/soccer/500/175.png"
3	http://football.org/ent/como	"Como"	"Red Cards"	"6"^^xsd:integer	"https://a.espncdn.com/i/teamlogos/soccer/500/2572.png"
4	http://football.org/ent/dortmund	"Borussia Dortmund"	"Red Cards"	"6"^^xsd:integer	"https://a.espncdn.com/i/teamlogos/soccer/500/124.png"
5	http://football.org/ent/mallorca	"Mallorca"	"Red Cards"	"6"^^xsd:integer	"https://a.espncdn.com/i/teamlogos/soccer/500/84.png"
6	http://football.org/ent/marseille	"Marseille"	"Red Cards"	"6"^^xsd:integer	"https://a.espncdn.com/i/teamlogos/soccer/500/176.png"

Figure 14: Get Top 10 Clubs By Stat (Output)

3.3 Graph

3.3.1 Get Entire Graph

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX fut-rel: <http://football.org/rel/>

SELECT ?subject ?predicate ?object ?subjectType ?objectType
```

```
WHERE {
    ?subject ?predicate ?object .

    # Get types
    OPTIONAL { ?subject rdf:type ?subjectType }
    OPTIONAL { ?object rdf:type ?objectType }

    # Filter out unrelated triples
    FILTER (!STRSTARTS(STR(?predicate), "http://www.w3.org/2000/01/rdf-schema"))
    FILTER (!STRSTARTS(STR(?predicate), "http://www.w3.org/1999/02/22-rdf-syntax-ns"))
    FILTER (!STRSTARTS(STR(?predicate), "http://proton.semanticweb.org/protonsys"))
    FILTER (!STRSTARTS(STR(?predicate), "http://football.org/stat"))
    FILTER (!STRSTARTS(STR(?subject), "http://football.org/stat"))
    FILTER (?predicate != <http://football.org/rel/photo_url>)
    FILTER (?predicate != <http://www.w3.org/2002/07/owl#inverseOf>)
}
```

	subject	predicate	object	subjectType	objectType
1	http://football.org/ent/luke_thomas	fut-rel:name	"Luke Thomas"	fut-rel:Player	
2	http://football.org/ent/nikola_maksimovic	fut-rel:name	"Nikola Maksimović"	fut-rel:Player	
3	http://football.org/ent/samuel_mbangula	fut-rel:name	"Samuel Mbangula"	fut-rel:Player	
4	http://football.org/ent/paul_nebel	fut-rel:name	"Paul Nebel"	fut-rel:Player	
5	http://football.org/ent/jesper_karlstrom	fut-rel:name	"Jesper Karlström"	fut-rel:Player	
6	http://football.org/ent/pietro_comuzzo	fut-rel:name	"Pietro Comuzzo"	fut-rel:Player	
7	http://football.org/ent/mads_hermansen	fut-rel:name	"Mads Hermansen"	fut-rel:Player	

Figure 15: Get Entire Graph (Output)

3.3.2 Get Specific Node And Adjacent Nodes (Example)

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX fut-rel: <http://football.org/rel/>

SELECT ?subject ?predicate ?object ?subjectType ?objectType
WHERE {
    # Get relationships where selected node is the subject
    {
        <http://football.org/ent/heidenheim> ?predicate ?object .
        BIND(<http://football.org/ent/heidenheim> AS ?subject)

        # Get types
        OPTIONAL { ?subject rdf:type ?subjectType }
        OPTIONAL { ?object rdf:type ?objectType }

        # Filter out non-relevant predicates
        FILTER (!STRSTARTS(STR(?predicate), "http://www.w3.org/2000/01/rdf-schema"))
        FILTER (!STRSTARTS(STR(?predicate), "http://www.w3.org/1999/02/22-rdf-syntax-ns"))
        FILTER (!STRSTARTS(STR(?predicate), "http://football.org/stat"))
        FILTER (?predicate != <http://football.org/rel/photo_url>)
    }
    UNION
    # Get relationships where selected node is the object
    {
        ?subject ?predicate <http://football.org/ent/heidenheim> .
        BIND(<http://football.org/ent/heidenheim> AS ?object)

        # Get types
        OPTIONAL { ?subject rdf:type ?subjectType }
        OPTIONAL { ?object rdf:type ?objectType }

        # Filter out non-relevant predicates
        FILTER (!STRSTARTS(STR(?predicate), "http://www.w3.org/2000/01/rdf-schema"))
        FILTER (!STRSTARTS(STR(?predicate), "http://www.w3.org/1999/02/22-rdf-syntax-ns"))
        FILTER (!STRSTARTS(STR(?predicate), "http://football.org/stat"))
        FILTER (?predicate != <http://football.org/rel/photo_url>)
    }
}
```

	subject	predicate	object	subjectType	objectType
1	http://football.org/ent/heidenheim	fut-rel:name	"1. FC Heidenheim 1846"	fut-rel:Club	
2	http://football.org/ent/heidenheim	fut-rel:city	"Heidenheim"	fut-rel:Club	
3	http://football.org/ent/heidenheim	fut-rel:stadium	"Voith-Arena"	fut-rel:Club	
4	http://football.org/ent/heidenheim	fut-rel:logo	"https://a.espncdn.com/i/teamlogos/soccer/500/6418.png"	fut-rel:Club	
5	http://football.org/ent/heidenheim	fut-rel:abrv	"HDH"	fut-rel:Club	
6	http://football.org/ent/heidenheim	fut-rel:country	http://football.org/ent/GER	fut-rel:Club	fut-rel:Country
7	http://football.org/ent/heidenheim	fut-rel:color	"DA0308"	fut-rel:Club	

Figure 16: Get Specific Node (Output)

4 UI Functionalities

4.1 Homepage - Top Players & Clubs

The main page, dedicated to "Top Players & Clubs" features a interactive display of the top-10 ranked players17 or clubs18 based on various statistics.

Each item showcases a entity (a player or a club) with key details such as an icon, name, informational text, and the current statistics that the entity holds. The top entity within the statistics group is clickable, leading to the individual profile page of that player or club. Below this, a list of sub-entities is shown. These sub-entities represent other notable players or clubs within the same statistics group.



Figure 17: Dashboard - Players



Figure 18: Dashboard - Clubs

4.2 Player List

The Player List page is designed to present a comprehensive and searchable list of football players. Each row is interactive, allowing users to click on a player entry to navigate directly to its dedicated page for further details. This page provides various filters, allowing users to search for players based on their name, position, club, and nation. This filtering system makes it easier to narrow down and find specific players that meet the user's criteria.

Additionally, a pagination system at the end of the table ensures seamless browsing for users exploring multiple pages of clubs. Navigation links allow movement between pages, with options to jump to the first or last page for efficiency.



Figure 19: Players - All



Figure 20: Players - Filtered By Manchester and Portugal

At the bottom of the page, there is a button to open a modal window, allowing users to add a new player. The modal includes a form where the user can input various details about the player, such as their name, positions, birth year, photo URL, nation, and the club they play for.



Figure 21: Players - Add Player

4.3 Player Details

The Player Details page provides an overview of a player's key information, including their name, nationality, birth details, age, clubs, and playing positions.

At the top, the player's name is prominently displayed, accompanied by a flag representing their country and their birth date. If available, the player's photo is shown on the left, ensuring quick visual recognition. The right side of the header displays the logo of the player's current club, with a link to the club's page.

Below the header, the page is divided into two main sections: Positions and Clubs. The Positions section lists all the roles the player can occupy on the field, each displayed as a labeled badge. Users can also add a new position through a modal form. The Clubs section showcases the teams the player has been part of, with their logos and names displayed in a structured list. The player's current club is highlighted distinctly. Users can add a new club association through a dedicated section.



Figure 22: Player Details

4.4 Club List

This page is similar to the players list: provides a comprehensive directory of football clubs, enabling users to explore and filter teams based on name and league. It features a clean and structured layout designed for easy navigation.

The main content of the page is presented in a structured table format, listing details for each club: name, logo, abbreviation, league, country (represented by a flag), and the number of players associated with the club. Pagination and filtering are also included.

Figure 23: Clubs List Filtered by Clubs that include "United"

4.5 Club Details

The Club Details page provides an in-depth overview of a specific football club, presenting essential information about the team, including its logo, location, stadium and league.

Below the header, the page lists the club's squad members in a table format. Each player's name, position, age, and country are displayed, with clickable links to individual player pages. This table is sortable, allowing users to sort the squad by name, position, age, or country. The table design is responsive and interactive, with a hover effect applied to each player row to improve the user experience.



Figure 24: Club Details (filtered by position)

4.6 RDF Graph

The RDF Graph Visualization page provides an interactive way to explore entities and their relationships in a structured graphical format as presented in figure 25. It allows users to dynamically view RDF triples and their connections in a visually intuitive manner. The page consists of three main sections: the graph visualization, the node details panel, and the RDF triples table. The central graph visualization, built using D3.js and the force-graph library, represents entities as nodes and their relationships as edges. Nodes are automatically color-coded based on their entity type, and directional arrows indicate the flow of relationships. Users can interact with the graph by clicking on nodes to view detailed information.

When a node is selected, the node details panel updates to display its label, type, and unique identifier. If the node represents a player or a club. Simultaneously, the RDF triples table updates dynamically, displaying the subject-predicate-object relationships related to the selected node. A toggle switch

allows users to enable or disable the display of human-friendly names for predicates instead of raw RDF URIs, improving readability.



Figure 25: RDF Graph Visualization

4.7 Player Connection & Comparison

In order to check attribute connections between football players, for instance, if they've played for the same club, came from the same country, or play in the same position, the following page (figure 26), allows us to choose two different players and verify if they link each other by the given criteria. In case of match, the attribute intersection will be shown, otherwise, will display a message warning that no match was found. It is also possible to compare the statistics between the players, organized by the various categories (figure 27). A green font indicates the better player for each particular stat.

5 Conclusion

This report presents the development of a football data web service aimed at providing comprehensive and interconnected information about clubs and players. By leveraging datasets from Kaggle and scraping data from zerozero.pt, a semantic network was constructed to enhance the accessibility and usability of football-related data.

The results demonstrate that the objectives were accomplished: the semantic network model constructed proved to be effective in showing the relationships



Figure 26: Player Connection Check



Figure 27: Player Comparison

between entities and the data associated and an intuitive interface was developed to demonstrate it.

We believe that this project allowed us to successfully apply the knowledge gained in class in a practical context. If we consider the project as a final product, we are quite satisfied with the outcome. The large variety of data and the attractive, accessible interface make it a potentially highly useful tool. However, to make the product even better and bigger, we would need additional data:

- Additional club and player statistics
- Additional relationships between new entities to enable more meaningful information about players and clubs
- Additional features, e.g., the ability to display top players within a specified team, league, or nationality based on selected statistics.

With these additions, the project could be an even more helpful resource for users looking for more detailed knowledge on the world of football.

6 Configurations To Run

In order to have the project in your workspace and run it locally, you can go to our GitHub repository at https://github.com/Sytuz/WS-Project-1, there you have all the necessary documentation in the README.md to setup the project.

Nevertheless, here are the steps to run our project:

- Unzip the folder or clone the repository using the following commands:

  ```
  git clone https://github.com/yourusername/WS-Project-1.git
  ```

- Then, you need to run:

  ```
  cd WS-Project-1
  docker compose up --build -d
  ```

After executing these commands, you can access our web application at http://localhost:8000 and the GraphDB's Workbench at http://localhost:7200/.

In rare occasions, an error can occur when accessing the web application, which shows an error screen (figure 28). This means that the repository is still being created at GraphDB and you will need wait one more minute or it was not successfully created due to formatting issues of a bash script, that could happen, while using Windows. If this happens, there are two possible solutions:

- **1.** Open the file $WS - Project - 1/data/config/init - repository.sh$ in a code editor such as Visual Studio Code and alter the 'End of Line Sequence' from **CRLF** to **LF** and save (if LF is already selected, select **CRLF**, save and then select **LF** once more). Once this is done, run:

  ```
  docker compose down -v
  docker compose up --build -d
  ```

- **2.** Manually create a repository named 'football' and load the RDF located at $WS - Project - 1/data/import/football - rdf - data - 3$.



Figure 28: Possible Error Page

References

[1] The most popular sports in the world - worldatlas, 2025. URL `https://www.worldatlas.com/articles/what-are-the-most-popular-sports-in-the-world.html`.

[2] Football players stats (2024-2025), 2025. URL `https://www.kaggle.com/datasets/hubertsidorowicz/football-players-stats-2024-2025`.

[3] Espn soccer data, 2025. URL https://www.kaggle.com/datasets/excel4soccer/espn-soccer-data.

[4] zerozero.pt :: Porque todos os jogos começam assim..., 2025. URL https://www.zerozero.pt/.